

COMPUTING FOURIER COEFFICIENTS AND FOURIER SERIES WITH MATHEMATICA

Below is a short Mathematica program that computes Fourier coefficients and series for any function that is $2L$ periodic and satisfies the Dirichlet conditions. Please go through this program carefully; it will help you learn many useful Mathematica and programming tools. I will imbed some comments in the program. Also, please remember Mathematica has a very good Documentation Center that you can access by clicking on "Help" on the Mathematica toolbar. The program below computes the Fourier series and Fourier coefficients for the step wise function used in Example 1, p. 353 of Boas.

The program :

```

In[58]:= Clear[function, limits, a0, a, b, fourierseries, variable, lowerlimit, upperlimit]
limits = {variable, lowerlimit, upperlimit};
(* The statement above allows me to
   define the array used in the 'Integrate' command. *)

variable = x; upperlimit =  $\pi$ ; lowerlimit =  $-\pi$ ;

(*I define the piecewise function, f(x) ,using the 'Which' command: *)

function[x_] := Which[ $-\pi < x < 0$ , 0,  $0 < x < \pi$ , 1]

(*The next three statements compute the Fourier coefficients. I can make evaluation of
   the coefficients easier indicating we are interested only in integer values of
   n. Refer to the Documentation Center to read about the 'Assumptions' command. Please
   see section after the output to learn how to make special symbols like  $\epsilon$ . *)

a0 = Integrate[function[x], limits] /  $\pi$ ;
a[n_] := a[n] = Integrate[function[x] Cos[n x], limits, Assumptions  $\rightarrow n \in \text{Integers}$ ] /  $\pi$ 

b[n_] := b[n] = Integrate[function[x] Sin[n x], limits, Assumptions  $\rightarrow n \in \text{Integers}$ ] /  $\pi$ 

(* Notice that I defined a[n] and b[n] as functions
   of n. Having had some experience with programs like this,
   I knew it would be important to make use of the 'trick' we learned in lab while
   investigating Fibonacci numbers. We have now computed the Fourier coefficients. *)

(* The statement below uses the computed coefficients to produce a Fourier series. *)

fourierseries = a0 / 2 + Sum[a[n] Cos[n x] + b[n] Sin[n x], {n, 1, 21}];

(* Now we produce output. The first output line will print
   out the explicit terms in the Fourier series out to n = 21. *)


Expand[fourierseries]

(* Next, we plot two curves on the same set of axes. I define the
   curves as the variables g1 and g2. The first curve is the plot of the
   Fourier series. The second curve is the plot of the original function
   (defined via a 'Which' statement). In order to distinguish the two overlapping curves,
   I use 'PlotStyle' to draw the piecewise function in a different font/format. Refer
   to the Documentation Center for more information on 'Tooltip'. *)

g1 = Plot[Tooltip[a0 / 2 + Sum[a[n] Cos[n z] + b[n] Sin[n z], {n, 1, 21}]], {z,  $-\pi$ ,  $\pi$ ]];
g2 = Plot[Tooltip[function[x]], {x,  $-\pi$ ,  $\pi$ }, PlotStyle  $\rightarrow$  {Thick, Red, Dashed}];
Show[g1, g2]

```

$$\begin{aligned} \text{Out}[66]= & \frac{1}{2} + \frac{2 \sin[x]}{\pi} + \frac{2 \sin[3 x]}{3 \pi} + \frac{2 \sin[5 x]}{5 \pi} + \frac{2 \sin[7 x]}{7 \pi} + \frac{2 \sin[9 x]}{9 \pi} + \\ & \frac{2 \sin[11 x]}{11 \pi} + \frac{2 \sin[13 x]}{13 \pi} + \frac{2 \sin[15 x]}{15 \pi} + \frac{2 \sin[17 x]}{17 \pi} + \frac{2 \sin[19 x]}{19 \pi} + \frac{2 \sin[21 x]}{21 \pi} \end{aligned}$$



Notes on creating special characters with Mathematica :

You would expect that a mathematical software platform would provide a fairly easy way to write the array of special symbols used in math and physics. Find the `ESC` key on your computer. You will be using it a lot.

One common pattern for most of these special symbols is `ESCcodeESC`

When you hit the `ESC` key on your computer, Mathematica will produce three vertical dots : `ESC` (That might be hard to see, but just hit the `ESC` key and see what happens.) Now if you want to type Greek letters, try typing `ESCaESC`. You should get lower case Greek α . You can experiment with both upper and lower cases.

Below are the codes for some of the mathematical symbols we will use frequently this semester :

Integration sign : `ESCintESC` $\rightarrow \int$

Partial derivative : `ESCpdESC` $\rightarrow \partial$

Del operator : `ESCdelESC` $\rightarrow \nabla$

Infinity : `ESCinfESC` $\rightarrow \infty$

Summation : `ESCsumESC` $\rightarrow \sum$

Pi : `ESCpESC` $\rightarrow \pi$

Element : `ESCelESC` $\rightarrow \in$

Escape key : `ESCescESC` $\rightarrow \text{ESC}$ (Make sure that you leave a space between the first `ESC` and the 'e')

Contour Integral : `ESCcintESC` $\rightarrow \oint$

From physics :

`ESC hbESC` $\rightarrow \hbar$ (hbar from quantum mechanics)

`ESC AngESC` $\rightarrow \text{\AA}$ (the Ångstrom unit of length)

Logical operators

Or : `ESC || ESC` $\rightarrow \vee$

And : `ESC && ESC` $\rightarrow \wedge$

For all : `ESC faESC` $\rightarrow \forall$

Exists : `ESC exESC` $\rightarrow \exists$

Another technique to produce these characters uses the format : `\[code]`

If you type in 'Euro' in the place of 'code' you get : €

Yen : ¥

And finally, some silly stuff :



If you type 'special characters' into the search bar on the documentation center, you will find many links to hundreds if not thousands of these symbols.